


MAKE YOUR COMMODORE 64


Sing

ED BOGAS



A CREATIVE PASTIMES BOOK





make your commodore 64® sing

ED BOGAS



Illustrated by Al Pagan

A Creative Pastimes Book
Reston Computer Group
Reston Publishing Company, Inc.
A Prentice-Hall Company
Reston, Virginia

ISBN 0-8359-4181-7

Copyright[©]1984 by Reston Publishing Company, Inc., A Prentice-Hall Company, Reston, Virginia 22090

All rights reserved. No part of this book may be reproduced in any way or by any means without permission in writing from the publisher.

Interior design and production by Laura Cleveland.

Commodore 64 is a registered trademark of Commodore Business Machines.

The page proofs for this book were composed on a TYXSET 1000 System by TYX Corporation in Reston, Virginia, and output using a Canon LBP-10 Laser Beam Printer. The camera-ready typeset pages were also composed on a TYXSET 1000 System and output using a Mergenthaler 101.

TYXSET 1000 is a trademark of TYX Corporation.

10 9 8 7 6 5 4 3 2 1

Printed in the United States of America.


contents

	<i>preface</i>	iv
CHAPTER 1	<i>in the beginning</i>	1
CHAPTER 2	<i>later in the beginning</i>	9
CHAPTER 3	<i>much later in the beginning</i>	17
CHAPTER 4	<i>near the end of the beginning</i>	25
CHAPTER 5	<i>you guessed it—the end of the beginning</i>	33
CHAPTER 6	<i>growing up</i>	43
CHAPTER 7	<i>extra-special delivery</i>	51
CHAPTER 8	<i>why me?</i>	59
CHAPTER 9	<i>why not?</i>	67
CHAPTER 10	<i>so there!</i> <i>epilogue</i>	77 89

***p*reface**


This book is an introduction to the uses of the Commodore 64 sound chip. You don't need a disk drive or a joystick for the rather short programs presented, and you don't have to know how to play an instrument or read music.


You *do* have to have a Commodore 64 and a video monitor or TV set. If you do, then I hope you enjoy the variety of tones, noises, and melodies contained in these pages. Even more, enjoy the ones you make up yourself!



CHAPTER 1

in the beginning





I AM HONORED TO WRITE THIS
biography although I have previously authored biographies of other great composers — Johann Sebastian BOX, Ludwig Van BYTEhoven, and Johannes bROMS (also pronounced bRAMS), as well as pieces on contemporary pop music idols such as Elvis PRESSKEY, the BITles, and Linda RUNstadt. I feel that giving the masses an inside look at the man behind the greatest music of all time has to be the highlight of my career. And so it thrills me to present a musical portrait (up close and personal) of Philip Joseph Roberts.


As source material for this biography I have drawn on interviews with his friends, teachers, and parents, as well as a diary he kept (and parts of which he has graciously agreed to have reproduced here). The diary serves as the main source of information about his early adulthood, at which time he kept mostly to himself.

One of the most surprising things about his childhood was how completely his talent was overlooked. This comment from his piano teacher, a Mrs. Foghat, was typical:

Yes, I vaguely remember P.J. He was a snotty little kid who thought playing the piano would be great, but decided early on that practicing was a drag. Pretty typical for a seven-year-old I'd say. He did have a cat that seemed to have some talent. That cat would run up and down the keyboard and actually play a little tune! What was his name? . . . oh yes, you wanted to know about P.J. Well, as I said, it's hard for me to imagine how he could have become

the world's greatest composer — the child could hardly read music ...

Obviously, P.J.'s unique gifts didn't emerge until long after he stopped taking lessons from Mrs. Foghat. Picking up where Mrs. Foghat gave up, let us look at this entry from P.J.'s diary, dated Christmas Day of his eighth year:



I got a Commodore 64 for Christmas and boy is it neat! It can even make music, and I can forget all about practicing!

When you first turn it on, the screen says COMMODORE 64 BASIC and a bunch of other stuff, then READY, and there's a blinking square. That's where the next thing you type will appear.

If you don't want to look at all that, you can hold down the SHIFT key and press the CLR HOME key; that clears the screen.

The way you make sounds on the computer is to "POKE" some numbers into some "addresses" in the computer's memory. For instance, the number in address 54296 tells the computer how loud to play. When you first turn on the computer, it has 0 at the address, which means the volume is "off." If you tune the TV set so that there isn't much hum and type

POKE 54296, 15

and press RETURN, you hear a little "pop." That's the computer tugging at the speaker of the TV set. To hear the "pop" again, type

POKE 54296, 0

and press **RETURN**. The computer makes a little pop as it "lets go" of the speaker. You can poke any value from 0 to 15 into address 54296. 0 is "off," 1 is very soft, 2 a little louder, and so on up to 15.

To make the computer play a note, you have to POKE some more values into some more addresses. Typing

POKE 54296, 15

and pressing **RETURN** (you have to press **RETURN** after each line you type so that the computer will carry out the instruction) sets the volume at loud. The computer also needs to know what "shape," what "pitch," and what "tone" to play. The book gives an example and says it will explain the addresses and values later. Anyway, I typed these lines:

POKE 54296, 15

POKE 54277, 143

POKE 54273, 40

POKE 54272, 40

POKE 54276, 17

I heard a note! To hear it again, I had to type

POKE 54276, 16

(to "turn off" the note) and then type


POKE 54276, 17

again. Pretty neat, huh, diary? I can't wait 'til the next chapter in the book so I can find out what all these numbers mean. So what if I can't play the piano as well as my cat — I bet I can do this computer stuff!







ACCEPTING MRS. FOGHAT'S OPINION AS
being fact that P.J. had no talent, his continuing to explore
new areas of music makes one realize that Philip Joseph
Roberts was either one heck of a stupid kid, or an incredible
genius. (In retrospect, of course, we realize the latter is
true.)



CHAPTER 2

later in the beginning






ALL THAT CHRISTMAS VACATION, WHILE other children were out playing football or soccer or in general just messing around, P.J. stayed glued to his new computer learning how to make it make sounds. One of P.J.'s boyhood friends, Peter, remembers:

It was really weird. I'd go by his house as usual and yell up, 'Hey, P.J., wanna mess around?' or 'Hey, P.J., there's a soccer game down at the park'; but he'd just lean out the window and say he was busy. After a while I stopped trying.

Of course, there was no way for Peter to have known what had P.J. so enthralled. From his diary we find out P.J. was excitedly working on developing a new facet of his computer.



Okay, diary, now I know more about all those numbers. Address 54296 controls the volume. The numbers you can POKE are from 0 (off) up to 15 (loudest). Address 54277 controls how the note is *attacked* and how it *decays* or "rings off." For instance, a big bell has a sharp attack and a long decay, a gong or the sound of plane going by has a slow attack and a long decay, and a knock on the door has a sharp attack and a fast decay. (They won't tell me which value does what yet, but I know the numbers range from 0 to 255.)

Addresses 54273 and 54272 tell the computer what *pitch* to play. They don't explain these numbers yet either, but they say that, basically, the higher the number the higher the pitch. These numbers also go from 0 to 255.

Finally, address 54276 controls the "tone" or "waveform" the computer will play. If you POKE 17 into 54276 it turns on a nice, smooth waveform. If you POKE 16 into it, it turns the waveform off.

Now I understand things a little better.

POKE 54296,15 sets the VOLUME high

POKE 54277,143 sets a MEDIUM ATTACK and a LONG DECAY

POKE 54273,40 and POKE 54272,40 sets the PITCH

POKE 54276,17 activates a nice, smooth waveform, and
 POKE 54276,16 turns it off

I found out something else, too. Computers run on "programs." A *program* is a list of instructions to the computer that are numbered so that the computer knows in what order to carry them out. Once you have a program typed in, you can "run" it by typing RUN and pressing **RETURN**. You can RUN the program again and again without retyping as long as the program hasn't been changed or erased from the memory. For instance, I typed in this program:

10 POKE 54296,15
 20 POKE 54296,0

Every time I typed RUN and pressed **RETURN**, I heard the speaker click on and off. I gotta find out more about this stuff.

IGNORING PETER CAMPBELL'S PRELIMINARY
remarks about what a show-off P.J. became, it is clear from the rest of Peter's recollections of a particular incident that, even at the tender age of eight, P.J. had a great desire to share his music:

One day we were all playing football and P.J. came running up. "Hey you guys," he yelled, "You gotta hear this!" We told him to shut up, it was the end of the fourth quarter and the score was tied, but he kept whining so we finally went back to his house with him. Up in his room he played some dumb noise on his computer. We all said, "So what?" and went back to our game. Unfortunately, it was too dark by then to finish, so we had to call it a tie. After that nobody talked to P.J. for about a week.

This event seemed to shake P.J.'s confidence in himself a little, as reflected in the following quote from his diary, and we can all be thankful that he didn't "bag it" at age eight:

7 7 7 7 7

I've learned a new way to start a program.

10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L



It's called a FOR/NEXT loop, and loops are very useful in computer programs. What it does when you RUN it is to set L=54272, then POKE 0 into that address, then the NEXT L sends the computer back to line 10, where it sets L=54273 (one larger), POKEs 0 into *that* address, goes NEXT L again, and so on. When L gets to be 54297, the computer sees that it's past the limit 54296, so it doesn't POKE anything but goes on with the rest of the program (if there is more). The reason the manual tells you to start every program this way is to "clear" all the sound addresses so that a program has a "clean start" without using values from earlier programs.

Anyway, I was working with this program:

10 FOR L=54272 TO 54296

20 POKE L, 0

30 NEXT L

(This clears the sound addresses)

40 POKE 54296, 15

(Sets the volume high)

50 POKE 54277, 10

(Sets a short ATTACK and DECAY)

60 POKE 54273, 32:POKE 54272, 32

(Sets the pitch. You can put these two commands on the same line if you separate them by a colon.)

70 POKE 54276, 129

(Sets the WAVEFORM to "noise.")

Every time I would RUN this program, I could hear a little "explosion." Then I added these lines:


```
80 POKE 54276,128  
90 GOTO 70
```

When I ran this, the computer went nuts! The only way I could stop it (short of turning it off) was to hold down the **RUN STOP** key. That's because line 80 turns the noise waveform off and then line 90 sends the computer back to line 70, which turns it on again. Holding down the **RUN STOP** key stops this "loop." I was afraid I had lost the program, but when I typed LIST and pressed **RETURN** I saw it was still there.

I had to show this program to somebody, so I made some of the guys come listen to it. But all they said was "So what?" Now they won't even talk to me. Gee, maybe I'm not any good at this either.

¢ ¢ ¢ ¢ ¢ ¢


IT TOOK HALF HIS LIFETIME, BUT THE
world says "So what?" no more to the work of Philip Joseph Roberts.



CHAPTER 3

much later in the beginning






AT THE AGE OF 10, P.J. WAS AN EXPERT AT
getting his computer to make sounds. His father, a janitor
at the local Pizza Shack, while very proud of his son, was
unable to afford lessons or advanced equipment for his son.
As he puts it:

*If I'd known he was going to spend all his time
playing with that computer, I never would have
bought it. I mean, I worked hard all day keep-
ing the "Shack" shiny, and you'd think when
I got home my son might have done a little
work around the yard for me, or at least got-
ten himself elected captain of the football team,
but no. So, when the little brat wanted more
equipment, I told him, "fat chance, kiddo, if
you want to spend somebody's money, make it
your own! Get out there and earn it like I do!"
Of course, I'm proud of him now, but he was a
pain when he was a kid.*

At 10 years old it was clear P.J. wasn't going to get a job driving a delivery van, so he could hardly believe his own good fortune when he found employment that spring. Once again we quote from his diary.



Mr. Deeler over at Kiddy-O Video Games says if I can program sound effects, he'll give me a job. I've got a couple of pretty good noises I could show him. Here's a gunshot.



(By the way, diary, since I'm writing so many programs these days I always type NEW and press RETURN first to erase old programs from the memory.)

```
10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L
40 POKE 54296,15
50 POKE 54277,12
60 POKE 54273,22:POKE 54272,132
70 POKE 54276,129
```

This program really makes the sound of a gunshot when you RUN it. The value 12 in line 50 gives the sound a sharp attack and a real long decay to 0.

Here's how I got it. The value you POKE in is the sum of two or more numbers, some for attack and some for decay.

The attack numbers are

- 128 (slow attack)
- 64 (quicker)
- 32 (quick)
- 0 (immediate)

The decay numbers are

- 8 (very long)
- 4 (shorter)
- 2 (short)
- 1 (real short)
- 0 (immediate)

You can add these numbers up in any combination to get different AD "envelopes" as the book calls them (attack-

decay envelopes). I used $0+8+4=12$ to get an immediate attack and an extra-long decay.

Now, here's a machine-gun program. It's similar to the gunshot program so instead of typing NEW and starting from scratch, I just changed the gunshot program by typing a few new lines to replace the old ones. First I changed line 50 for a shorter decay:

50 POK 54277,9

Then I added a line to turn the noise waveform off:

80 POK 54276,128

Then I added a "loop" statement:

100 GOTO 70

This is just like the program I showed my "former" friends. To stop it once I had started RUNning it, I had to press the **RUN STOP** key.

To make it sound more like a machine gun, I put in a "pause" line:

90 FOR T=1 TO 50:NEXT T

This line makes the computer wait until it counts up to 50 before it turns the noise back on. It sounds just like a machine gun when you run it. To keep from having to stop the program by pressing the **RUN STOP** key, I put in one more loop.


65 FOR G=1 TO 10

100 NEXT G

Try running that, diary!

AS COULD BE EXPECTED, P.J. NERVOUSLY
but bravely took his little bag of explosions and noises to
show Mr. Deeler.

After trying to locate Deeler for several weeks — checking aliases, bankruptcies, and fictitious addresses to no avail — we gave up. However, the following excerpts from P.J.'s diary give us a pretty good idea of Mr. Deeler's character (or lack thereof) without a personal interview.



I got the job! Mr. Deeler gave me a graphics program from one of his games and told me to put sounds to it!

```
100 FOR N=0 TO 24
110 POKE 1054+40*N,81
120 POKE 55326+40*N,1
130 POKE 1054+40*N,32
140 NEXT N
```

If you RUN it, a ball falls down.

To add sound to the program, these are the lines I used:

```
10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L
40 POKE 54296,15
50 POKE 54277,123
60 POKE 54272,60
105 POKE 54273,60-N
```

115 POKE 54276,17
135 POKE 54276,16

Lines 10-30, of course, clear the sound addresses. Line 40 sets the volume at 15 (loudest). Line 50 sets the AD envelope to quick attack, long decay and line 60 sets the lower-address pitch value to 60. Line 105 is pretty tricky. As N gets bigger, the pitch value 60-N gets smaller, so the pitch gets lower as the ball falls.

Then I used the loop to make sound get softer as the ball falls. First I typed

40

(and pressed **RETURN**) which erases line 40. Then I added


102 POKE 54296, 15-N/2

to the program. When N=0, $15-N/2=15$, which is the highest volume. When N=24, $15-N/2=3$, which is pretty soft.

I really like this job. Mr. Deeler is swell. And here's the best part! Instead of just a dumb salary, Mr. Deeler's going to give me a share in the company!

HE CLOSES THIS ENTRY IN HIS DIARY with a comment that shows he hasn't forgotten his friends in his first rush of success.

"So what?", huh? I'll show them!"



CHAPTER 4

*near the end of the
beginning*





DISREGARDING PETER CAMPBELL'S OBSER-
vation that if given an inch, P.J. would take a mile, it is true
that his work for Mr. Deeler seemed to help P.J. regain
some of his shaken confidence. He even seemed like an
average kid for a while, playing soccer and football with his
friends.

They all still liked to mess around, but individual tastes
began to change; it came as a surprise to everybody when
Peter Campbell started taking violin lessons. However, it
seemed he had quite a talent.

P.J.'s interests and tastes began to shift a little, too. He
found himself attracted to a girl at school. Annie Smart was
— smart, that is — and was also considered a rather talented
dancer. Unfortunately, Annie was equally attracted — to
Peter Campbell.

Never one to fade in the face of adversity, P.J. tried hard
to attract her attention and somehow gain her affection. His
best bet, he decided, was to do something wonderful on the
computer.



I bet if I really learn this "envelope" stuff I could make my Commodore 64 sound like a real instrument, say a clarinet. It turns out that besides the AD envelope, there's another envelope called sustain/release (or SR for short). Here's what it does.



The AD envelope raises the note to the maximum volume and then "decays" it down. So far, it's decayed down to 0, but if we set the sustain to something above 0, that's where it will decay to. Once the waveform has been turned off, it will release down to 0.

The address for this is 54278. The numbers for *sustain* are

128 LOUD
64 MEDIUM
32 LOW
16 LOWEST
0 NONE

For *release* they are

8 VERY LONG
4 MEDIUM
2 SHORT
1 VERY SHORT
0 IMMEDIATE

Just as with the AD envelope, you can add these numbers together to get different SR effects.

One more thing: If you put a delay statement in between the "waveform on" and "waveform off" statements, the computer will hold the note at the SUSTAIN level and not go into release until it gets to the "waveform off" statement. That's what line 40 in the next program does.

This program makes a real pretty tone, sort of like a clarinet. The AD value in line 60 is 128+8, for a medium attack and a long decay. The SR value in line 70 is 128+7+4+2+1, for a loud sustain and a very long release.

```
10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L
40 POKE 54296,15
50 POKE 54273,32:POKE 54272,32
60 POKE 54277,136
70 POKE 54278,143
80 POKE 54276,17
90 FOR N=1 TO 2000:NEXT N
100 POKE 54276,16
```

Pretty, huh? By the way, to STOP a program with a sustain in it, the **RUN STOP** key may not work. Hold down the **RUN STOP** key and press **RESTORE**, which POKEs 0's into a lot of addresses.

♪ ♪ ♪ ♪ ♪ ♪

IN THE MEANTIME, PETER WAS BECOMING a school celebrity because of his violin playing, and on Friday afternoons gave little concerts. Naturally, then, it was a Friday afternoon when P.J. finished his prize program and could hardly wait to invite Annie over to hear it. Annie was polite and apologetic but definitely was not going to miss Peter in concert to listen instead to whatever it was P.J. did on his computer.

P.J. was not immune to the sting of rejection but, given his own budding genius (and what Peter Campbell unkindly refers to as his huge ego, which we will — with more tolerance and forbearance due to our realization of how

extraordinarily gifted P.J. was even then — refer to as "the need for an artist to have his work acknowledged"), P.J. simply had to have an audience for his first melody. He writes about it in his diary.



I guess I'll never learn. I got my three buddies to listen to this program, the *very first* melody program I ever wrote:

```
10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L
40 POKE 54296,15
44 READ A,B
46 IF A=0 THEN STOP
50 POKE 54273,A:POKE 54272,B
60 POKE 54277,136
70 POKE 54278,143
80 POKE 54276,17
90 FOR N=1 TO 1000:NEXT N
95 POKE 54276,16
100 GOTO 44
200 DATA 32,32,40,150,54,54,48,68
210 DATA 42,192,36,36,24,29,0,0
```

Line 44, a READ statement, tells the computer to look at the first DATA line (line 200) and let A and B take the first two values (32 and 32). Line 50 POKEs these values into the pitch addresses. Lines 60 and 70 are a clarinet envelope.

Then lines 80 to 95 play the note, hold it, and release it.


Line 100 sends the computer back to the READ statement, where it looks at the *next two* values (40 and 150) and uses them for pitch values, and so on.

Finally, to stop the process, I put 0,0 and the end of DATA line 210. Line 46 is an IF/THEN statement: IF something is true THEN the computer will do something else, in this case STOP.

I ran this program for my three buddies, and do you know what they said? I'll tell you what they said. Two of them said "Got anything to eat?" and the third one said, "So what?" That's what they said.

• • • • •


IF IT IS TRUE THAT FOR ARTISTS TO BE
truly great they must suffer, what was to happen next
probably helps account for P.J.'s great musical triumphs.



CHAPTER 5

*you guessed it — the end of
the beginning*





P.J. WORKED FOR MR. DEELER FOR ABOUT
two years. Every Wednesday he would bring him his latest
videogame sounds, and Mr. Deeler would give him a new
assignment. If nothing else, P.J. got a lot of satisfaction
from his work.



I made a real interesting discovery today! I was working on
a bouncing ball program for Mr. Deeler.

```
80 READ A,B,C
100 FOR N=A TO B STEP C
110 POKE 1054+40*N,81
120 POKE 55326+40*N,1
140 POKE 1054+40*N,32
150 NEXT N
155 IF C=-1 THEN RESTORE
160 GOTO 80
200 DATA 0,24,1,24,0,-1
```

This isn't the discovery, but there are a couple of new things in this program. First, FOR/NEXT loops don't have to count by ones. They can count by twos, or threes, or anything, even backwards (by -1). You specify the counting "step" by adding a STEP command (STEP 3, for instance, means count by threes). So, in line 80 the READ command gets three values and plugs them into line 100. The first three values (0,24,1) make the ball fall. The next three (24,0,-1) make it go back up.

To keep it going, line 155 gives the RESTORE command, which makes the READ command take values from the *beginning* of the data line. The IF condition is true when the ball is at the top. The RESTORE makes it fall again. To stop the ball, you press the **RUN STOP** key.

The same sound ideas work for this program. You add these lines:

```
10 FOR L=54272 TO 54296
20 POKE L,0
30 NEXT L
40 POKE 54296,15
50 POKE 54277,127
60 POKE 54272,60
105 POKE 54273,60-N
115 POKE 54276,17
135 POKE 54276,16
```

I ran this program and was watching the ball bounce. I was thinking that the ball should make a sound when it hits an edge. Too bad the voice is busy.

That's when I made my discovery! I hadn't seen it in the manual before. The Commodore 64 has another voice!

The important addresses are:

AD:54284
SR:54285
PITCH:54280 and 54279
WAVEFORM:54283

I decided to put my gunshot in the program at the top and bottom. All I had to do was add these lines:

156 POKE 54284, 12

(Second voice AD)

157 POKE 54280, 22

158 POKE 54279, 132

(Second voice pitches)

159 POKE 54283, 129


(Turns on the noise waveform)

152 POKE 54283, 128

(Turns the noise waveform off)


Mr. Deeler really liked it!

HOWEVER, EVEN THOUGH P.J. HAD PLENTY of what Mick Jagger couldn't get any of ("Satisfaction"), he didn't get paid any money at all for his efforts. Mr. Deeler had assured P.J. he would compensate him with stock in Kiddy-O Video, and that if he had to translate that into dollars and cents, P.J. would be set for life, or at least until he was 21! Being a trusting soul, and (we're forced to agree with Peter Campbell's analysis here) just a little dumb, P.J. had believed him. It is obvious from an oblique reference in his diary to finding Kiddy-O deserted and Mr. Deeler long-since gone one Wednesday, that P.J. shared Peter's analysis of himself as well. We asked P.J.'s mother about this calamity, and she said (in part):



Well, you can imagine, can't you? The look on his face that afternoon when he came back from the store, and it was all boarded up and Mr. Deeler vanished just like that? I thought he was sick; P.J. always looked like death warmed over when he got the least bit of a cold — he was a very colicky baby, and ... what?, Oh well, P.J. told me he wasn't sick, just dumb. Well, I always had thought he was — well, you know — not the brightest little thing in the world, but I didn't know it showed on your face like the flu ... what?, Oh well, I'm sure old Mr. Deeler is kicking himself now for having been so sleazy — what with P.J. turning out to be the world's greatest composer and all ... although how could he have known? I mean I'm his mother, and it sure came as a shock to me ... what? Oh, well. That's all I can remember.

However dedicated and optimistic P.J. continued to be, glory days of recognition were still far off, and the following humiliation, which he recorded in his diary, could have been what drove him temporarily into retreat.



There's a third voice on the computer (I looked ahead)! The addresses are:

AD:54291
SR:54292

PITCH:54287 and 54286
WAVEFORM:54290

This gave me the idea of adding one more element to Mr. Deeler's program. Holding down the **SHIFT** key and pressing **CLR HOME** clears the screen. You can put this instruction into the bouncing ball program by typing

90 PRINT "hold down the **SHIFT** key and press **CLR HOME** }"

What appears in the parentheses is a heart shape but, when the computer gets to line 90, it clears the screen. Next I added

112 POKE 1519+N,82
113 POKE 55791+N,3
144 POKE 1519+N,32

When I would RUN this program, a dash would move across the screen. To add sound (using the *third* voice) I added these lines:

70 POKE 54291,7

(Immediate attack, medium decay)

75 POKE 54286,60:POKE 54287,16

(Pitch)

103 POKE 54290,33

(Turn on buzzy waveform)

148 POKE 54290,32

(Turns off waveform)

I tried to LIST the program before I ran it, but it was too long for the screen. To see part of it, say lines 10 to 100, you type LIST 10-100. Anyway I ran it and it looked great. So I decided to enter it in the Talent Contest at school today. They've got a big monitor and a sound system in the auditorium. Maybe this'll show 'em I've got talent!

* * * * *


AND THEN, THAT SAME EVENING, P.J.
made the following entry in his diary.


#####

Peter Campbell won the Talent Contest with a violin solo. Annie Smart took second with a jazz dance. I guess that's no surprise. Guess who won third prize?

When I came out to plug my Commodore into the auditorium system, my cat, who had followed me to school, jumped on the piano and ran up and down the keys. Everybody started laughing, and they gave *her* the third prize! I never even got to run my program.


THIS IS THE LAST ENTRY IN P.J.'S DIARY
for almost four years.





CHAPTER 6

growing up






DURING THE REST OF P.J.'S SCHOOL YEARS he seemed to make a conscious effort to blend into the crowd. His high school yearbook depicts him as an average student participating in an average number of average activities. Only the hobby listed under his senior picture gives any clue that P.J. may not have been as average as the average student. It says: "Hobbies: writes computer software."


His father says,

Yeah, P.J. kinda straightened out when he was a teenager. He wasn't quarterback of the varsity football team or anything, and forgot about him picking up the yard, but he got a real job after school and got paid for it, and he used his own money for real stuff like football equipment. He was a pretty good kid. He never quit messing around with that computer, though — I think his mother kept some of the stuff he wrote — want to see it?

Following is a rare example of P.J.'s early published work.



As you all know, the pitch of a note is determined by two numbers, for instance, the two POKEd into addresses 54273 and 54272. Both numbers range from 1 to 255, and the higher the number, the higher the pitch for each one. But



there is a big difference in the way the two numbers work, as the following games will show you. In the first, you try to guess the higher-address value:

```
10 FOR L=54272 TO 54296:POKE L,0:NEXT L
20 POKE 54296,15
30 POKE 54277,135:POKE 54278,135
40 POKE 54273,33:POKE 54272,65
60 POKE 54276,17
70 FOR T=1 TO 1500:NEXT T
75 POKE 54276,16
77 IF B=33 THEN GOTO 200
80 PRINT "TYPE IN YOUR GUESS FOR THE"
90 PRINT "HIGH NUMBER AND PRESS RETURN"
95 PRINT "YOU MAY GUESS FROM 1 TO 255"
100 INPUT B
110 PRINT "YOUR NOTE SOUNDS LIKE THIS:"
120 POKE 54273,B:POKE 54276,17
130 FOR T=1 TO 1500:NEXT T
135 POKE 54276,16
140 PRINT "MINE SOUNDS LIKE THIS:"
150 GOTO 10
200 PRINT "YOU WIN!"
```

The statement 100 INPUT B waits for you to make a guess (and hit **RETURN**), then 120 POKEs that value into address 54273 and plays the note. Most people can tell if their guess is different from the computer's, and can eventually find the value. (If your guess sounds too high, lower the guess. If it's too low, raise it.) If you don't want to finish, you can stop the program by holding down **RUN STOP** and pressing **RESTORE**.

Now try this game, where you try to guess the lower-address value. Change these lines in the first game and then RUN the program:

```
77 IF B=65 THEN GOTO 200  
90 PRINT "LOW NUMBER AND PRESS RETURN"  
120 POKE 54272,B:POKE 54276,17
```

It's much harder to hear differences in this game because the lower-address value makes a "fine-tuning" adjustment. You might try this game on friends who claim to have a great ear for music! In some cases the difference can't be heard at all.



THIS GAME WAS NOT POPULAR FOR THREE reasons. First, the second (lower address) value was too hard to guess. Second, all you had to do to get the answer was LIST the program, and third, once you had solved it, it lost all its interest. But P.J. improved it greatly by doing away with second number guessing and making the computer choose the first number at *random*, or unpredictably. (For more about random numbers, see Michael Orkin's excellent book *Random Alley Adventure*, Reston Publishing). All he had to do was change and add these few lines:


```
35 X=INT(240*RND(1))+10  
40 POKE 54273,X:POKE 54272,65  
77 IF B=X THEN GOTO 200
```


```
90 PRINT "1ST NUMBER AND PRESS RETURN"
120 POKE 54273,B: POKE 54276,17
150 GOTO 40
```

This game was (very probably) different from one RUN to the next and since the number RND(1) is an unpredictable decimal between 0 and 1, X is an unpredictable number between 10 and 249.

The game tested well and the company P.J. wrote it for thought it would sell.

Unfortunately, that year, when P.J. was 19, was not the best for software sales, or anything else, because that was the year the aliens landed.






CHAPTER 7

extra-special delivery






ANTICIPATION IS ALWAYS BETTER THAN
realization, so after an initial panic, the landing of the aliens was put into perspective worldwide as "not nearly so exciting an event as we had always imagined it would be." However, the aliens did play a crucial part in Philip Joseph Roberts' life, so we shall take the time to explain parts of what happened that summer.

The visionary gifts of Jules Verne were confirmed that summer, as the spaceships themselves (hundreds of them landing simultaneously all over the planet) looked remarkably like descriptions he had written nearly a hundred years earlier. The ships defied all systems of defense, acknowledged no hailing signals, took no action, and made no response. They just landed and sat there. After about six months, they all took off simultaneously, relocated themselves, and landed again. They continued to repeat the process at approximately six-month intervals and, as the months went by the people of Earth began to relax and accept the ships. Small installations of militia were stationed around each craft, and vendors dotted the perimeters of each installation supplying the inevitable tourists with hot dogs and I SAW THE ALIENS bumper stickers.

After months of public speculation and innumerable scientific colloquia, the people of Earth had no clearer idea of the reasons behind the "invasion" than the day the aliens had landed. One thing seemed apparent: Whatever they were up to, they meant us no harm.

So, then, we have the scenario of world events as P.J. resumed writing in his diary, by which time he was 19 years old.





Well, it looks like I've got a choice to make: To finish college or start working full time right now. The Mystery Programs are starting to sell pretty well especially in Canada. I'll show one to you, diary.

It's a lot like that tone-guessing program I did before but, instead of just saying "YOU WIN," it spells out a Mystery Message! People seem to like that. To hide the message, I put it in number code:

```
2 FOR K=1 TO 8
5 X=INT(240*RND(1))+10
10 FOR L=54272 TO 54296:POKE L,0=NEXT L
20 POKE 54296,15
30 POKE 54277,135:POKE 54278,135
40 POKE 54273,X:POKE 54272,65
60 POKE 54276,17
70 FOR T=1 TO 1500:NEXT T
75 POKE 54276,16
76 IF X=B THEN GOTO 200
88 PRINT "GUESS HIGH NUMBER":INPUT B
90 IF B < 1 THEN 88
95 IF B >255 THEN 88
120 POKE 54273,B
130 PRINT "YOUR NOTE SOUNDS LIKE THIS:"
140 POKE 54276,17
150 FOR T=1 TO 1500:NEXT T
155 POKE 54276,16
160 PRINT "MINE SOUNDS LIKE THIS:"
170 GOTO 10
200 RESTORE
210 PRINT "YOU NOW HAVE";K;"CORRECT"
215 PRINT:PRINT
```

```
220 FOR J=1 TO K
230 READ M:PRINT CHR$(M);
240 NEXT J
245 PRINT:PRINT:PRINT
260 NEXT K
270 PRINT "YOU WIN, HOSER!"
500 DATA 84,65,75,69,32,79,70,70
```

If you play the whole game, it goes through the loop from 2 to 260 eight times. Each time you win, the loop from 220 to 240 reads one more letter from the data line. (The READ statement starts from the beginning each time because of the RESTORE command at 200.)

The function CHR\$ changes the number code to a letter or space and eventually prints out the entire eight-character secret message.


The company has put out a whole line of these "Mystery Messages" programs — in birthday greetings, Christmas cards, and so on. The biggest seller is an 18-character message "GO BACK TO THE ZOO."

Things have been OK, but it looks like I'm not going to "show 'em" after all. I got a letter from Peter; he's studying in Europe this summer. Annie Smart is in New York now, dancing on Broadway. Hooray for them.

I've met a nice girl. Her name is Wanda Wye, and she's a programmer too. We went to see the aliens last Sunday. Maybe I'll see her again sometime. Oh, yeah, one other thing. I got a mysterious letter today in my mailbox. There was no postmark on it, and inside the envelope was a program that said SOLVE ME on the top. The weird thing is that it's just like the program I just showed you with these changes:

```
2 FOR K=1 TO 17
270 PRINT "BE"
500 DATA 87,69,32,65,82,69,32
510 DATA 84,72,69,32
520 DATA 65,76,73,69,78,83
```


I'll try it tomorrow. If you solve it first, diary, let know!



CHAPTER 8

why me?






WANDA WYE LATER BECAME MRS. PHILIP

Joseph Roberts (or Wanda Wye-Roberts, as she prefers) and during our interview she fondly remembered an early romantic encounter with the world's greatest composer:

P.J. and I worked at the same software company. He was going to school part time, I remember, and seemed very nice, if somewhat continually distracted. He had always been so low-key (sometimes I wondered if he had blood pressure) that I was truly amazed when he came into work one day actually looking excited. I must admit to a little disappointment though, when I realized it was not me exciting him; whatever it was, he wanted to share it with me, and that was a giant step in the right direction. It seemed that something strange had happened to him and, as soon as he had figured it out, he wanted me to know all about it.



Dear Diary: I solved that puzzle. The message is: "WE ARE THE ALIENS." Who could have sent me a message like that? I know it has to be a prank, so why am I so excited about it?

Well, I'd better get to work. I'm working on a new game — a memory test game. First I get the 1, 2, 3, and 4 keys to stand for four pitches. Here it is:

```
10 FOR L=54272 TO 54296:POKE L,0:NEXT L  
20 POKE 54296,15  
30 POKE 54277,7
```



```
40 GET A$: IF A$ = "" THEN 40
50 IF A$ = "1" THEN 400
60 IF A$ = "2" THEN 400
70 IF A$ = "3" THEN 400
80 IF A$ = "4" THEN 400
85 GOTO 40
400 A = VAL(A$) * 30
500 POKE 54273, A: POKE 54272, A
510 POKE 54276, 17: POKE 54276, 16
520 GOTO 40
```

(You have to press **RUN STOP** to stop this program.) The GET command in line 40 is like INPUT except that it doesn't wait; it takes the first key you hit as the value of A\$. If you don't press anything (so that A\$ = ""), the program stays at line 40. If you don't press 1, 2, 3, or 4, line 85 sends the program back to line 40.

If you do press 1, 2, 3, or 4, the program goes to line 400 and POKEs values into the pitch address. These values are 30 times "VAL(A\$)"(which may be 1, 2, 3, or 4), so the values are 30, 60, 90, or 120.

To make a memory test game out of it, just limit our playing to four notes with these two statements:

```
38 FOR K = 1 TO 4
520 NEXT K
```

Now the program will stop after the player plays four notes.

Next we get the computer to choose (at random) a sequence of length four from the set (1, 2, 3, 4).

```
2 FOR B = 1 TO 4
```

```
4 T(B)=30*INT(RND(1)*4+1)
6 NEXT B
```

T(B) is an example of an array T(1), T(2), etc. Arrays make it easier to handle and compare lists of data.

Next we get the computer to play its four choices in order:

```
32 FOR J=1 TO 4
33 POKE 54273,T(J):POKE 54272,T(J)
34 POKE 54276,17:POKE 54276,16
35 FOR T=1 TO 300:NEXT T:NEXT J
```

Finally, we give instructions to the player, in the play booklet, to copy what the computer plays. The computer compares its choices with the player's guesses and decides whether to say "YOU WIN" or "TRY AGAIN."

```
515 C(K)=30*VAL(A$)
600 R=0
610 FOR W=1 TO 4
620 IF T(W)=C(W) THEN R=R+1
630 NEXT W
640 IF R=4 THEN 700
650 PRINT "TRY AGAIN"
660 FOR T=1 TO 500:NEXT T:GOTO 32
700 PRINT "YOU WIN!"
```

Line 515 makes an "array" out of the player's guesses. The variable R decides whether the player wins or not. Line 600 sets R to 0. Line 620 adds 1 if a guess is correct. Line 640 sends the player the reward only if *all four* guesses are right.

To make the game harder, the upper limits in lines 2, 32, 38, and 610 and the reward value R in line 640 can be set as high as 10.

FOLLOWING IS AN ENTRY FROM LATER
that week.


I got another puzzle in the morning mail. The note with it said, "Reproduce these nine tones and you will get a secret message." The weird thing is, the program is just like the program I just showed you but with these differences:


```
2 FOR B=1 TO 9
32 FOR J=1 TO 9
38 FOR K=1 TO 9
610 FOR W=1 TO 9
640 IF R=9 THEN 700
700 FOR S=1 TO 22
710 READ M:PRINT CHR$(M);
720 NEXT S
800 DATA 66,69,32,72,79,77,69,32
810 DATA 84,79,78,73,71,72,84,32
820 DATA 65,78,68,32,66,69
```

AND, EVIDENTLY, A FEW MOMENTS LATER:

♪ ♪ ♪ ♪ ♪


Me?! Why me??!






CHAPTER 9

why not?





THE THIN LINE BETWEEN GENIUS AND
madness is easily crossed. It is even, according to some other biographers of great artists, routinely crossed. Bearing this consensus in mind, it is easy for us to now accept the incredible diary entries P.J. made next as a necessary developmental stage of a blossoming genius. Then, however, Wanda could not be blamed for only seeing P.J.'s behavior as very strange. She elaborates:

P.J. dragged me aside one day at work and whispered in my ear. To tell you the truth, he could have said just about anything and I would have been happy to listen. I had fallen in love with the guy. Anyway, he whispered in my ear that he had been contacted by the aliens, and that they were to contact him at home that night, and would I please be there with him. I used to give points for originality when it came to lines, so he got a 10 for that one, but watching his expression while he continued to whisper to me, I realized he was utterly serious.


P.J.'s diary fills us in on how that evening began.




Wanda and I were sitting on the sofa. I think she wanted to hold hands, but my palms were sweaty so I kind of avoided contact. We hadn't been there very long when a disembodied



voice announced that I was to go alone to Buena Park the next night for further instructions.



NEITHER WANDA NOR P.J. WOULD DISCUSS
that next night. P.J., however, referred me to his diary.



As I wait to go to see the aliens, I have an urge to compose a piece of music that expresses the way I am feeling. After all these years of only composing music for games, I hardly know where to begin.

My heart is beating loudly and chills are going up and down my spine. Maybe the sound of my heartbeat would be a good start for my music.

So far the only waveforms I've used have been TRIANGLE (ON 17, OFF 16), a smooth sound; SAWTOOTH (ON 33, OFF 32), a rough sound; and NOISE (ON 129, OFF 128). There is one waveform I've never used, I guess because it's a little more complicated, but it'd be good for a heartbeat. It's called PULSE and its code is ON 65, OFF 64. It also needs two "pulse width" values which go into addresses 54275 and 54274. The high-address value can go from 0 to 15, and the lower from 0 to 255.

To me this sounds like a heartbeat:

```
5 FOR C=1 TO 100
10 FOR L=54272 TO 54296:POKE L,0:NEXT L
20 POKE 54296,15
30 POKE 54277,7
```

```
35 POKE 54275,1:POKE 54274,10
37 POKE 54273,3:POKE 54272,3
40 POKE 54276,65
50 FOR N=1 TO 500:NEXT N
55 POKE 54276,64
60 NEXT C
```

As I think about the aliens, and reflect, my heart beats faster. To capture this excitement in my music, I can make the program cycle faster with a few changes and additions. (Line 44 keeps it steady once it gets fast.)

```
42 IF C<20 THEN D=C
44 IF C>19 THEN D=20
50 FOR N=1 TO 500-D*20:NEXT N
```

Now I want to put something in that gives the feeling of a shiver up the spine. I think that gunshot sound I did long time ago will work, but I'll have to POKE it into the second voice addresses (AD at 54284, pitch at 54280 and 54279, waveform 54283).

I want to put the gunshot in between the pulses, so I'll divide the delay of line 50 into two half-length delays and insert the gunshot. I'll have to lose the little pick-up to the pulse, though, because going back to line 10 and shutting everything off interrupts the heartbeats.

```
5
15 FOR C=1 TO 100
```

(That moves the loop line past the "clear" (line 10))

25 POKE 54284, 12

(Sets the gunshot AD)

38 POKE 54280, 22 : POKE 54279, 132

(The gunshot pitches)

50 FOR N=1 TO 250-10 *D:NEXT N

(Half the pause)

52 IF INT(C/4) <> C/4 THEN GOTO 55

(The gunshot sounds every fourth time)

54 POKE 54283, 128 : POKE 54283, 129

(Turns the gunshot off, then on)

58 FOR N=1 TO 250-10 *D:NEXT N

That's it! Just how I feel — scared and excited. My timing's not bad, either; I have just enough of it left to call Wanda before I have to leave.

AT THIS POINT I MUST INTERJECT A PERSONAL opinion, and at the same time ask you, the reader, to accept it. I feel strongly that Mr. Roberts is the world's greatest composer — an opinion shared, I might add, by the leading authorities of the music world. However

widespread this opinion might be, and I assuredly include the reader, the fact remains that the following excerpt from P.J.'s diary might tend to undermine this truth, and I plead with you to play a recording of one of his works just now before you read the next portion of this. This is simply to remind you of his great genius and to reaffirm your own personal respect for a man who will undoubtedly sound to you now like a lunatic.

♪ ♪ ♪ ♪ ♪ ♪

The Buena Park alien installation was relatively deserted. The usual soldiers were on guard and a few tourists were milling around. At exactly the appointed time, one of the tourists came up to me and said:

"What's happening, P.J.?"

"Excuse me," I responded, "Who are you?" The tourist gestured toward the alien ships.

"Oh!" I gasped. "But ... you look like us!"

"*You are descended from us,*" said the alien.

"Who?" I stammered, "Me?"

"All of you," said the alien. "Look, didn't you ever think it was peculiar that, of all the creatures on this planet, *only* human beings have written language or a sense of history?"

"Well, uh, aren't we just more highly developed ...?"

"If you were more highly developed, you would have keener senses, quicker bodies, and so on. You are *differently* developed, because you are *not from* this planet."

"Then where are we from?" I asked. The alien paused a moment to look around — we were alone, so he continued.



"Thousands of years ago, our race embarked on a project of writing the history of the Galaxy. We dispatched hundreds of expeditions, each with the necessary technology to adapt their bodies to survive in alien worlds. Earth was one of these worlds."

"Occasionally some scouts would adapt too well and succeeding generations would lose track of their purpose. That's what happened here."

"So you've been adapted, too," I said.

The alien nodded. "In order to come here, yes."

"And what is your mission?"

"To find out what happened to the original expedition. Now we know. And we know there's no going back for you all. You're Earthlings now, through and through."

"Why are you telling *me* all this?" I asked.


"Because there is a second part to my mission," replied the alien. "We are *still* writing the history of the Galaxy. So we need someone here to transmit useful information back to us — magazines, films, tapes — it's easy. I'll show you how."

"Me? You picked *me* to do your transmissions? Why?"

"Why not?" the alien shrugged. "You're as good a choice as anyone. Besides, I like those Mystery Messages of yours. Anyway we're not asking you to do this for free. You'll be compensated."

"Compensated? How?"


The alien smiled, then walked away. In my head I heard the words, "So what?"



CHAPTER 10

so there!






TUCKED BETWEEN THE PAGES OF P.J.'S
diary, we came across a letter postmarked a week after the entry we just noted. The letter was from the Music Museum in New York and was addressed to P.J. The letter itself was short and to the point. The Hon. J. Hunt Hogshat, III, Curator of the Music Museum, was commissioning a computer-generated piece of music from P.J., anticipating its performance on International Stereo-TV later that year. He also requested a sample of P.J.'s work.

During our interview with Mr. Hogshat, we requested an explanation of his intuitive solicitation from Philip Joseph Roberts, P.J. being totally unknown and unheralded at that time. It is an interesting enough aside to note here that perhaps the people responsible for discovering and introducing artists and their work to the public share that same delicate mental balance with the artists. How else could we rationalize Mr. Hogshat's total lack of recollection as to how he found out about P.J. in the first place? Bearing this shared mental stability problem in mind, we return to P.J.'s diary.



So *this* is the compensation. They're going to plant the suggestion in people's minds that I'm hot stuff. Well, maybe I am! One thing's for sure; I'm going to give it my best shot! Let's see, what shall I write? I'd better get out my keyboard values.

Here is a full piano keyboard octave. The Commodore 64 plays eight octaves, from the lowest (octave 0) to the highest (octave 7).



Here is the table that gives the high and low address values for each note of each octave.

	OCTAVE								
	0	1	2	3	4	5	6	7	
HIGH	C	1	2	4	8	16	33	67	134
LOW		12	24	48	97	195	135	15	30
HIGH	C#	1	2	4	8	17	35	71	142
LOW		28	56	112	225	195	134	12	24
HIGH	D	1	2	4	9	18	37	75	150
LOW		45	90	180	104	209	162	69	139
HIGH	D#	1	2	4	9	19	39	79	159
LOW		62	125	251	247	239	223	191	126
HIGH	E	1	2	5	10	21	42	84	168
LOW		81	163	71	143	31	62	125	250
N	HIGH	F	1	2	5	11	22	44	89
O	LOW		102	204	152	48	96	193	131
T	HIGH	F#	1	2	5	11	23	47	94
E	LOW		123	246	237	218	181	107	214
HIGH	G	1	3	6	12	25	50	100	200
LOW		145	35	71	143	30	60	121	243
HIGH	G#	1	3	6	13	26	53	106	212
LOW		169	83	167	78	156	57	115	230
HIGH	A	1	3	7	14	28	56	112	225
LOW		195	134	12	24	49	99	199	143
HIGH	A#	1	3	7	14	29	59	119	238
LOW		221	187	119	239	223	190	124	248
HIGH	B	1	3	7	15	31	63	126	253
LOW		250	244	233	210	165	75	151	46

My First Symphony. First I'll clear the sound addresses and set the volume!

```
10 FOR L=54272 TO 54296:POKE L,0:NEXT L  
20 POKE 54296,15
```

Then I'll choose some notes — C, G, and A from the third octave and G, A, B from the second octave. I'll put the values in two arrays so they'll be easy to get to.

```
100 A(0)=8:B(0)=97:A(1)=12:B(1)=143  
110 A(2)=14:B(2)=24:A(3)=6:B(3)=71  
120 A(4)=7:B(4)=12:A(5)=7:B(5)=233
```

Okay, now I want to do something symphonic. Maybe I'll begin with a low repeated figure that will make people wonder what's coming next.

I'll try to play this line repeatedly:



(Boy, I barely remember how to read music!)

I'll set the AD to immediate attack, long decay, and then write a loop to play it. This is fun!

```
200 POKE 54277,63  
210 FOR N=1 TO 16  
220 READ M  
230 POKE 54273,A(M):POKE 54272,B(M)  
240 POKE 54276,17
```

```
250 FOR T=1 TO 60:NEXT T
260 POKE 54276,16
300 NEXT N
310 RESTORE:GOTO 210
500 DATA 0,1,2,1,3,1,2,1,0,1,2,1
510 DATA 3,1,4,5
```

It sounds like a rumbling clarinet section. It would be nice to have a harmony line in the second voice. I'll use the same data, but I'll choose D,E,F,C,A from the fourth octave and C from the fifth so I can play this:



```
130 C(0)=25:D(0)=30:C(1)=21:D(1)=31
140 C(2)=18:D(2)=209:C(3)=33:D(3)=135
150 C(4)=22:D(4)=96:C(5)=28:D(5)=49
200 POKE 54277,63:POKE 54284,63
235 POKE 54280,C(M):POKE 54279,D(M)
240 POKE 54276,17:POKE 54283,17
260 POKE 54276,16:POKE 54283,16
```

Sounds pretty good! Now it needs a high melody line. I'll use C, E, G, and A from the sixth octave.

```
160 E(0)=100:F(0)=121:E(1)=84:F(1)=125
```

```
170 E(2)=67:F(2)=15:E(3)=112:F(3)=199
204 FOR J=0 TO 3
205 POKE 54291,16:POKE 54292,79
207 POKE 54287,E(J):POKE 54286,F(J)
208 POKE 54290,33
310 RESTORE:NEXT J
320 GOTO 204
```

Hmm. Not bad. Interesting how the rhythm gets a little "hitch" in it when the J-loop goes back to line 204. I guess that's the extra time it takes the computer to carry out those instructions. I'll have to fix that somehow. Because of the sustain in the third voice, it takes **RUN STOP** and **RESTORE** to stop the program.

Well, I guess that's good enough for a sample. I'll put one more loop in it to fade down the volume and end the program and then I'll send it off!

```
202 FOR V=15 TO 0 STEP -3
203 POKE 54296,V
320 NEXT V
```

THIS IS, OF COURSE, P.J.'S FIRST SYM-
phony, which drew rave reviews from every country in the world. It alone would have secured his reputation, but it was followed quickly by more works; each week he seemed to have a new one ready.

His diary reveals a continued overbearing modesty, though his mental balance seems to be improving.





I can't believe it! They took that sample I sent them, with the hitch in it, and performed it, and now everybody thinks I'm great! Those aliens really did a job on everybody. Peter Campbell writes me saying he wants me to write a violin concerto for him. Annie Smart wants a ballet suite!

I've got a feeling I could send in *anything* and they'd think it's great. I'll test that — I'll just change the last program a little bit to make it sound like a banjo and harmonica and see what they say. First I'll get rid of voice 2:

130
140
150
200 POKE 54277, 63
225
240 POKE 54276, 17
260 POKE 54276, 16

Ok, now I'll change the speed:

250 FOR T=1 TO 120: NEXT T

Next I'll change the AD envelope on the first voice to immediate attack, medium decay:

200 POKE 54277, 7

Now I'll change the waveform so that it sounds more banjo-like:

240 POKE 54276, 33
260 POKE 54276, 32


To give the high voice a more harmonica-like sound, I'll lengthen the attack and decay, eliminate the sustain, and turn the note on and off:


205 POK 54291,207
305 POK 54290,32

Now I'll put a little "swing" in the rhythm:

242 IF INT(N/2)=N/2 THEN X=0
244 IF INT(N/2)<>N/2 THEN X=100
250 FOR T=1 TO 120+X:NEXT T


I'll leave the hitch. They liked it last time.





epilogue





I LAST SAW PHILIP JOSEPH ROBERTS JUST before this book went to press. I was visiting his modest home in the mountains which he shares with his wife Wanda and their two children. He still speaks a little strangely, but now it's in a joking way, as if there were some story that only he and Wanda know.

"They fixed me good," he said as he waved a copy of National Geographic over a strange-looking ornamental mirror. "When I was a kid nobody took me seriously. Now I could play 'Chopsticks' and bring the house down. Success, it seems, is the end of all hope. I'm not better than I was then, I'm just the same. But you know what? That's OK with me.

"Besides, I've got a way out. Their hypnotism, or whatever it was, only applied to people who were alive then. Watch this. Jason, come over here."

P.J.'s six-year-old son approached.

"Type this program into the computer," said his father. Jason typed in P.J.'s famous "Toy Keyboard" program:

```
10 FOR L=54272 TO 54296:POKE L,0:NEXT L
20 POKE 54296,15
100 A(0)=8:B(0)=97:A(1)=10:B(1)=143
110 A(2)=6:B(2)=71:A(3)=A(1):B(3)=B(1)
200 POKE 54277,7
205 POKE 54284,7
210 FOR N=0 TO 3
212 GET A$: IF A$="" THEN 230
214 POKE 54280,ASC(A$):POKE 54279,ASC(A$)
216 POKE 54283,17
```

```
230 POKE 54273,A(N):POKE 54272,B(N)
240 POKE 54276,33
250 FOR T=1 TO 150: NEXT T
260 POKE 54276,32
270 POKE 54283,16
300 NEXT N
305 GOTO 210
```

The program plays a pattern while the ASC function in line 214 converts any key struck by the player into a number which is then POKED into the pitch addresses of Voice 2 (line 230).

P.J. RAN the program and played along with it, pressing different keys. After about a minute he held down the **RUN STOP** key and pressed **RESTORE**, stopping the program.


His melodies had been marvelous, and I said as much.

"Yes," sighed P.J., "I played the same thing at Carnegie Hall last month and the critics went wild. Jason, how did *you* like it?"

"Not much," said Jason.

P.J. laughed, a deep, relaxed sound. "I didn't like it much myself," he said.

And off they went, chasing the cat.



MAKE YOUR COMMODORE 64® SING

ED BOGAS

Here is a delightful introduction to the Commodore 64® home computer and the unique Commodore sound chip. **MAKE YOUR COMMODORE 64 SING** documents the exciting career of the famous computer composer P.J. Roberts—from the arrival of his first Commodore 64 to his composition of renowned computer-generated symphonies. As you read about P.J.'s struggles in becoming accepted as a true musician, you'll discover how the Commodore 64 creates its vast range of sound and music, and how to use the sound chip to make your own musical compositions.

You'll be with P.J. when he first learns to POKE values into the computer's addresses to create individual notes, different pitches, short and long duration of sounds, and volume levels. Like P.J., you'll figure out how to write short programs that will string a series of individual musical components together to make a tune or effect that you can play over and over. With the Commodore 64 you'll discover a stunning reality—you don't need to practice to play music perfectly! All you need is to know how to write programs!

MAKE YOUR COMMODORE 64 SING is packed with short programs that demonstrate each new effect, along with tips and explanations on why the effects work. You'll even find some programs that are puzzles—and, in order to make some sense of P.J.'s trials, you'll have to key in the programs on your own Commodore for the answer. Like P.J., you'll become comfortable with the different sound-producing elements, and you'll be able to combine them to create your own realistic sound effects, songs, and stirring symphonies in two- and three-part harmony. With this book, you'll not only learn about programming and music, you'll be having fun at the same time!

A Creative Pastimes Book
Reston Publishing Company, Inc.
A Prentice-Hall Company
Reston, Virginia



4

